

## Grammar for the ERPLAG Language

<program> ::= <moduleDeclarations><otherModules><driverModule><otherModules>

<moduleDeclarations> ::= <moduleDeclaration><moduleDeclarations> | ε

<moduleDeclaration> ::= DECLARE MODULE ID SEMICOL

<otherModules> ::= <module><otherModules> | ε

<driverModule> ::= DRIVERDEF DRIVER PROGRAM DRIVERENDDEF <moduleDef>

<module> ::= DEF MODULE ID ENDDEF TAKES INPUT SQBO <input\_plist> SQBC SEMICOL <ret> <moduleDef>

<ret> ::= RETURNS SQBO <output\_plist> SQBC SEMICOL | ε

<input\_plist> ::= ID COLON <dataType> <sub\_input\_plist>

<sub\_input\_plist> ::= COMMA ID COLON <dataType> <sub\_input\_plist> | ε

<output\_plist> ::= ID COLON <type> <sub\_output\_plist>

<sub\_output\_plist> ::= COMMA ID COLON <type> <sub\_output\_plist> | ε

<dataType> ::= INTEGER | REAL | BOOLEAN | ARRAY SQBO <dynamic\_range> SQBC OF <type>

<dynamic\_range> ::= <index> RANGEOP <index>

<type> ::= INTEGER | REAL | BOOLEAN

<moduleDef> ::= START <statements> END

<statements> ::= <statement> <statements> | ε

<statement> ::= <ioStmt> | <simpleStmt> | <declareStmt> | <conditionalStmt> | <iterativeStmt>

<ioStmt> ::= GET\_VALUE BO ID BC SEMICOL | PRINT BO <extended\_var> BC SEMICOL

<extended\_var> ::= <var> | TRUE | FALSE

<var> ::= ID <whichId> | NUM | RNUM

<whichId> ::= SQBO <index> SQBC | ε

<simpleStmt> ::= <assignmentStmt> | <moduleReuseStmt>

<assignmentStmt> ::= ID <whichStmt>

<whichStmt> ::= <lvalueIDStmt> | <lvalueARRStmt>

<lvalueIDStmt> ::= ASSIGNOP <expression> SEMICOL

<lvalueARRStmt> ::= SQBO <index> SQBC ASSIGNOP <expression> SEMICOL

<index> ::= NUM | ID

<moduleReuseStmt> ::= <optional> USE MODULE ID WITH PARAMETERS <idList> SEMICOL

<optional> ::= SQBO <idList> SQBC ASSIGNOP | ε

<idList> ::= ID <sub\_idList>

<sub\_idList> ::= COMMA ID <sub\_idList> | ε

<expression> ::= <arithmeticExpr> | <booleanExpr>

<arithmeticExpr> ::= <sub\_arithmeticExpr> <term>

<sub\_arithmeticExpr> ::= <term> <op1> <sub\_arithmeticExpr> | ε

<term> ::= <sub\_term> <factor>

<sub\_term> ::= <factor> <op2> <sub\_term> | ε

<factor> ::= <op1> <sub\_factor> | <sub\_factor>

<sub\_factor> ::= BO <arithmeticExpr> BC | <var>

<op1> ::= PLUS | MINUS

<op2> ::= MUL | DIV

<booleanExpr> ::= <sub\_booleanExpr> <logicalOp> <booleanExpr> | <sub\_booleanExpr>

<sub\_booleanExpr> ::= <arithmeticExpr> <relationalOp> <arithmeticExpr> | BO <booleanExpr> BC | TRUE | FALSE

<logicalOp> ::= AND | OR

<relationalOp> ::= LT | LE | GT | GE | EQ | NE

<declareStmt> ::= DECLARE <idList> COLON <dataType> SEMICOL

<conditionalStmt> ::= SWITCH BO ID BC START <caseStmt> <default> END

<caseStmt> ::= CASE <value> COLON <statements> BREAK SEMICOL <nullableCaseStmt>

<nullableCaseStmt> ::= CASE <value> COLON <statements> BREAK SEMICOL <nullableCaseStmt> | ε

<value> ::= NUM | TRUE | FALSE

<default> ::= DEFAULT COLON <statements> BREAK SEMICOL | ε

<iterativeStmt> ::= FOR BO ID IN <range> BC START <statements> END |  
WHILE BO <booleanExpr> BC START <statements> END

<range> ::= NUM RANGEOP NUM